NAG Library Routine Document

D03PXF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D03PXF calculates a numerical flux function using an Exact Riemann Solver for the Euler equations in conservative form. It is designed primarily for use with the upwind discretization schemes D03PFF, D03PLF or D03PSF, but may also be applicable to other conservative upwind schemes requiring numerical flux functions.

2 Specification

```
SUBROUTINE D03PXF (ULEFT, URIGHT, GAMMA, TOL, NITER, FLUX, IFAIL)

INTEGER            NITER, IFAIL
REAL (KIND=nag_wp) ULEFT(3), URIGHT(3), GAMMA, TOL, FLUX(3)
```

3 Description

D03PXF calculates a numerical flux function at a single spatial point using an Exact Riemann Solver (see Toro (1996) and Toro (1989)) for the Euler equations (for a perfect gas) in conservative form. You must supply the *left* and *right* solution values at the point where the numerical flux is required, i.e., the initial left and right states of the Riemann problem defined below. In D03PFF, D03PLF and D03PSF, the left and right solution values are derived automatically from the solution values at adjacent spatial points and supplied to the subroutine argument NUMFLX from which you may call D03PXF.

The Euler equations for a perfect gas in conservative form are:

$$\frac{\partial U}{\partial t}+\frac{\partial F}{\partial x}=0, \tag{1}$$

with

$$U=\begin{bmatrix}\rho\\ m\\ e\end{bmatrix} \quad \text{and} \quad F=\begin{bmatrix} m \\ \frac{m^2}{\rho}+(\gamma-1)\left(e-\frac{m^2}{2\rho}\right) \\ \frac{me}{\rho}+\frac{m}{\rho}(\gamma-1)\left(e-\frac{m^2}{2\rho}\right)\end{bmatrix}, \tag{2}$$

where ρ is the density, m is the momentum, e is the specific total energy and γ is the (constant) ratio of specific heats. The pressure p is given by

$$p=(\gamma-1)\left(e-\frac{\rho u^2}{2}\right), \tag{3}$$

where $u=m/\rho$ is the velocity.

The routine calculates the numerical flux function $F(U_L,U_R)=F(U^*(U_L,U_R))$, where $U=U_L$ and $U=U_R$ are the left and right solution values, and $U^*(U_L,U_R)$ is the intermediate state $\omega(0)$ arising from the similarity solution $U(y,t)=\omega(y/t)$ of the Riemann problem defined by

$$\frac{\partial U}{\partial t}+\frac{\partial F}{\partial y}=0, \tag{4}$$

with U and F as in (2), and initial piecewise constant values $U=U_L$ for $y<0$ and $U=U_R$ for $y>0$. The spatial domain is $-\infty<y<\infty$, where $y=0$ is the point at which the numerical flux is required.

The algorithm is termed an Exact Riemann Solver although it does in fact calculate an approximate solution to a true Riemann problem, as opposed to an Approximate Riemann Solver which involves some form of alternative modelling of the Riemann problem. The approximation part of the Exact Riemann Solver is a Newton–Raphson iterative procedure to calculate the pressure, and you must supply a tolerance TOL and a maximum number of iterations NITER. Default values for these parameters can be chosen.

A solution cannot be found by this routine if there is a vacuum state in the Riemann problem (loosely characterised by zero density), or if such a state is generated by the interaction of two non-vacuum data states. In this case a Riemann solver which can handle vacuum states has to be used (see Toro (1996)).

4 References

Toro E F (1989) A weighted average flux method for hyperbolic conservation laws *Proc. Roy. Soc. Lond.* **A423** 401–418

Toro E F (1996) *Riemann Solvers and Upwind Methods for Fluid Dynamics* Springer–Verlag

5 Parameters

1: ULEFT(3) – REAL (KIND=nag_wp) array *Input*

On entry: ULEFT(i) must contain the left value of the component U_i, for $i = 1, 2, 3$. That is, ULEFT(1) must contain the left value of ρ, ULEFT(2) must contain the left value of m and ULEFT(3) must contain the left value of e.

2: URIGHT(3) – REAL (KIND=nag_wp) array *Input*

On entry: URIGHT(i) must contain the right value of the component U_i, for $i = 1, 2, 3$. That is, URIGHT(1) must contain the right value of ρ, URIGHT(2) must contain the right value of m and URIGHT(3) must contain the right value of e.

3: GAMMA – REAL (KIND=nag_wp) *Input*

On entry: the ratio of specific heats, γ.

Constraint: GAMMA > 0.0.

4: TOL – REAL (KIND=nag_wp) *Input*

On entry: the tolerance to be used in the Newton–Raphson procedure to calculate the pressure. If TOL is set to zero then the default value of 1.0×10^{-6} is used.

Constraint: TOL ≥ 0.0.

5: NITER – INTEGER *Input*

On entry: the maximum number of Newton–Raphson iterations allowed. If NITER is set to zero then the default value of 20 is used.

Constraint: NITER ≥ 0.

6: FLUX(3) – REAL (KIND=nag_wp) array *Output*

On exit: FLUX(i) contains the numerical flux component $\hat{F}_i$, for $i = 1, 2, 3$.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the

recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

Note: if the left and/or right values of ρ or p (from (3)) are found to be negative, then the routine will terminate with an error exit (IFAIL = 2). If the routine is being called from the NUMFLX etc., then a **soft fail** option (IFAIL = 1 or −1) is recommended so that a recalculation of the current time step can be forced using the NUMFLX parameter IRES (see D03PFF or D03PLF).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, GAMMA $\leq$ 0.0,
or TOL $<$ 0.0,
or NITER $<$ 0.

IFAIL = 2

On entry, the left and/or right density or derived pressure value is less than 0.0.

IFAIL = 3

A vacuum condition has been detected therefore a solution cannot be found using this routine. You are advised to check your problem formulation.

IFAIL = 4

The internal Newton–Raphson iterative procedure used to solve for the pressure has failed to converge. The value of TOL or NITER may be too small, but if the problem persists try an Approximate Riemann Solver (D03PUF, D03PVF or D03PWF).

7 Accuracy

The algorithm is exact apart from the calculation of the pressure which uses a Newton–Raphson iterative procedure, the accuracy of which is controlled by the parameter TOL. In some cases the initial guess for the Newton–Raphson procedure is exact and no further iterations are required.

8 Further Comments

D03PXF must only be used to calculate the numerical flux for the Euler equations in exactly the form given by (2), with ULEFT(i) and URIGHT(i) containing the left and right values of ρ, m and e, for $i = 1, 2, 3$, respectively.

For some problems the routine may fail or be highly inefficient in comparison with an Approximate Riemann Solver (e.g., D03PUF, D03PVF or D03PWF). Hence it is advisable to try more than one Riemann solver and to compare the performance and the results.

The time taken by the routine is independent of all input parameters other than TOL.

9 Example

This example uses D03PLF and D03PXF to solve the Euler equations in the domain $0 \leq x \leq 1$ for $0 < t \leq 0.035$ with initial conditions for the primitive variables $\rho(x,t)$, $u(x,t)$ and $p(x,t)$ given by

$$\rho(x,0) = 5.99924, \quad u(x,0) = \quad 19.5975, \quad p(x,0) = 460.894, \qquad \text{for } x < 0.5,$$
$$\rho(x,0) = 5.99242, \quad u(x,0) = -6.19633, \quad p(x,0) = \quad 46.095, \qquad \text{for } x > 0.5.$$

This test problem is taken from Toro (1996) and its solution represents the collision of two strong shocks travelling in opposite directions, consisting of a left facing shock (travelling slowly to the right), a right travelling contact discontinuity and a right travelling shock wave. There is an exact solution to this problem (see Toro (1996)) but the calculation is lengthy and has therefore been omitted.

9.1 Program Text

```
!   D03PXF Example Program Text
!   Mark 24 Release. NAG Copyright 2012.

    Module d03pxfe_mod

!     D03PXF Example Program Module:
!            Parameters and User-defined Routines

!     .. Use Statements ..
      Use nag_library, Only: nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Real (Kind=nag_wp), Parameter        :: alpha_l = 460.894_nag_wp
      Real (Kind=nag_wp), Parameter        :: alpha_r = 46.095_nag_wp
      Real (Kind=nag_wp), Parameter        :: beta_l = 19.5975_nag_wp
      Real (Kind=nag_wp), Parameter        :: beta_r = 6.19633_nag_wp
      Real (Kind=nag_wp), Parameter        :: half = 0.5_nag_wp
      Integer, Parameter                   :: itrace = 0, ncode = 0, nin = 5,  &
                                              nout = 6, npde = 3, nxi = 0
!     .. Local Scalars ..
      Real (Kind=nag_wp)                   :: el0, er0, gamma, rl0, rr0, ul0,  &
                                              ur0
    Contains
      Subroutine bndary(npde,npts,t,x,u,ncode,v,vdot,ibnd,g,ires)

!       .. Scalar Arguments ..
        Real (Kind=nag_wp), Intent (In)      :: t
        Integer, Intent (In)                 :: ibnd, ncode, npde, npts
        Integer, Intent (Inout)              :: ires
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (Out)     :: g(npde)
        Real (Kind=nag_wp), Intent (In)      :: u(npde,npts), v(ncode),        &
                                                vdot(ncode), x(npts)
!       .. Executable Statements ..
        If (ibnd==0) Then
          g(1) = u(1,1) - rl0
          g(2) = u(2,1) - ul0
          g(3) = u(3,1) - el0
        Else
          g(1) = u(1,npts) - rr0
          g(2) = u(2,npts) - ur0
          g(3) = u(3,npts) - er0
        End If
        Return
      End Subroutine bndary
      Subroutine numflx(npde,t,x,ncode,v,uleft,uright,flux,ires)

!       .. Use Statements ..
        Use nag_library, Only: d03pxf
!       .. Scalar Arguments ..
        Real (Kind=nag_wp), Intent (In)      :: t, x
        Integer, Intent (Inout)              :: ires
        Integer, Intent (In)                 :: ncode, npde
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (Out)     :: flux(npde)
        Real (Kind=nag_wp), Intent (In)      :: uleft(npde), uright(npde),     &
```

```
                                          v(ncode)
!       .. Local Scalars ..
        Real (Kind=nag_wp)                :: tol
        Integer                           :: ifail, niter
!       .. Executable Statements ..
        tol = 0.0_nag_wp
        niter = 0

        ifail = 0
        Call d03pxf(uleft,uright,gamma,tol,niter,flux,ifail)

        Return
      End Subroutine numflx
    End Module d03pxfe_mod
    Program d03pxfe

!     D03PXF Example Main Program

!     .. Use Statements ..
      Use nag_library, Only: d03pek, d03plf, d03plp, nag_wp
      Use d03pxfe_mod, Only: alpha_l, alpha_r, beta_l, beta_r, bndary, el0,    &
                             er0, gamma, half, itrace, ncode, nin, nout, npde, &
                             numflx, nxi, rl0, rr0, ul0, ur0
!     .. Implicit None Statement ..
      Implicit None
!     .. Local Scalars ..
      Real (Kind=nag_wp)                  :: d, p, tout, ts, v
      Integer                             :: i, ifail, ind, itask, itol, k,   &
                                             lenode, mlu, neqn, niw, npts,    &
                                             nw, nwkres
      Character (1)                       :: laopt, norm
!     .. Local Arrays ..
      Real (Kind=nag_wp)                  :: algopt(30), atol(1), rtol(1),    &
                                             ue(3,9), xi(1)
      Real (Kind=nag_wp), Allocatable     :: u(:,:), w(:), x(:)
      Integer, Allocatable                :: iw(:)
!     .. Intrinsic Procedures ..
      Intrinsic                           :: real
!     .. Executable Statements ..
      Write (nout,*) 'D03PXF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) npts

      nwkres = npde*(2*npts+3*npde+32) + 7*npts + 4
      mlu = 3*npde - 1
      neqn = npde*npts + ncode
      niw = neqn + 24
      lenode = 9*neqn + 50
      nw = (3*mlu+1)*neqn + nwkres + lenode
      Allocate (u(npde,npts),w(nw),x(npts),iw(niw))

      Read (nin,*) gamma, rl0, rr0, ul0, ur0

      el0 = alpha_l/(gamma-1.0_nag_wp) + half*rl0*beta_l**2
      er0 = alpha_r/(gamma-1.0_nag_wp) + half*rr0*beta_r**2

!     Initialise mesh
      Do i = 1, npts
        x(i) = real(i-1,kind=nag_wp)/real(npts-1,kind=nag_wp)
      End Do
      xi(1) = 0.0_nag_wp

!     Initial values
      Do i = 1, npts
        If (x(i)<half) Then
          u(1,i) = rl0
          u(2,i) = ul0
          u(3,i) = el0
        Else If (x(i)==half) Then
          u(1,i) = half*(rl0+rr0)
```

```
          u(2,i) = half*(ul0+ur0)
          u(3,i) = half*(el0+er0)
        Else
          u(1,i) = rr0
          u(2,i) = ur0
          u(3,i) = er0
        End If
      End Do

      Read (nin,*) itol
      Read (nin,*) norm
      Read (nin,*) atol(1), rtol(1)
      Read (nin,*) laopt

      ind = 0
      itask = 1
      algopt(1:30) = 0.0_nag_wp

!     Theta integration
      algopt(1) = 2.0_nag_wp
      algopt(6) = 2.0_nag_wp
      algopt(7) = 2.0_nag_wp

!     Max. time step
      algopt(13) = 0.5E-2_nag_wp

      ts = 0.0_nag_wp
      tout = 0.035_nag_wp

!     ifail: behaviour on error exit
!            =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call d03plf(npde,ts,tout,d03plp,numflx,bndary,u,npts,x,ncode,d03pek,nxi, &
        xi,neqn,rtol,atol,itol,norm,laopt,algopt,w,nw,iw,niw,itask,itrace,ind, &
        ifail)

      Write (nout,99998) ts
      Write (nout,99999)

!     Read exact data at output points

      Do i = 1, 9
        Read (nin,*) ue(1:3,i)
      End Do

!     Calculate density, velocity and pressure

      k = 0
      Do i = 15, npts - 14, 14
        d = u(1,i)
        v = u(2,i)/d
        p = d*(gamma-1.0_nag_wp)*(u(3,i)/d-half*v**2)
        k = k + 1
        Write (nout,99996) x(i), d, ue(1,k), v, ue(2,k), p, ue(3,k)
      End Do

      Write (nout,99997) iw(1), iw(2), iw(3), iw(5)

99999 Format (4X,'X',7X,'APPROX D',3X,'EXACT D',4X,'APPROX V',3X,'EXAC','T V', &
        4X,'APPROX P',3X,'EXACT P')
99998 Format (/' T = ',F6.3/)
99997 Format (/' Number of integration steps in time = ',I6/' Number ', &
        'of function evaluations = ',I6/' Number of Jacobian ', &
        'evaluations =',I6/' Number of iterations = ',I6)
99996 Format (1X,E9.2,6(E11.4))
    End Program d03pxfe
```

9.2 Program Data

```
D03PXF Example Program Data
 141                                        : npts
 1.4  5.99924         5.99242
      1.175701059E2  -3.71310118186E1       : gamma, rl0, rr0, ul0, ur0
 1                                          : itol
 '2'                                        : norm
 0.5E-2  0.5E-3                             : atol(1), rtol(1)
 'B'                                        : laopt
 0.5999E+01   0.1960E+02   0.4609E+03
 0.5999E+01   0.1960E+02   0.4609E+03
 0.5999E+01   0.1960E+02   0.4609E+03
 0.5999E+01   0.1960E+02   0.4609E+03
 0.5999E+01   0.1960E+02   0.4609E+03
 0.1428E+02   0.8690E+01   0.1692E+04
 0.1428E+02   0.8690E+01   0.1692E+04
 0.1428E+02   0.8690E+01   0.1692E+04
 0.3104E+02   0.8690E+01   0.1692E+04  : ue
```

9.3 Program Results

```
 D03PXF Example Program Results

 T =  0.035

    X       APPROX D   EXACT D    APPROX V   EXACT V    APPROX P   EXACT P
 0.10E+00 0.5999E+01 0.5999E+01 0.1960E+02 0.1960E+02 0.4609E+03 0.4609E+03
 0.20E+00 0.5999E+01 0.5999E+01 0.1960E+02 0.1960E+02 0.4609E+03 0.4609E+03
 0.30E+00 0.5999E+01 0.5999E+01 0.1960E+02 0.1960E+02 0.4609E+03 0.4609E+03
 0.40E+00 0.5999E+01 0.5999E+01 0.1960E+02 0.1960E+02 0.4609E+03 0.4609E+03
 0.50E+00 0.5999E+01 0.5999E+01 0.1960E+02 0.1960E+02 0.4609E+03 0.4609E+03
 0.60E+00 0.1423E+02 0.1428E+02 0.8660E+01 0.8690E+01 0.1688E+04 0.1692E+04
 0.70E+00 0.1425E+02 0.1428E+02 0.8672E+01 0.8690E+01 0.1688E+04 0.1692E+04
 0.80E+00 0.1921E+02 0.1428E+02 0.8674E+01 0.8690E+01 0.1689E+04 0.1692E+04
 0.90E+00 0.3100E+02 0.3104E+02 0.8675E+01 0.8690E+01 0.1687E+04 0.1692E+04

 Number of integration steps in time =    697
 Number of function evaluations =   1708
 Number of Jacobian evaluations =     1
 Number of iterations =      2
```

Example Program 1
Euler Equation Solution Showing Collision of Two Strong Shocks
DENSITY



Euler Equation Solution Showing Collision of Two Strong Shocks
VELOCITY



Euler Equation Solution Showing Collision of Two Strong Shocks
PRESSURE